
１質点減衰系の地震応答―RESP 
 
 
 プログラム RESP（Response of Single Degree-of-Freedom System）は，地動加速度の時刻歴が
与えられたとき，1 質点減衰系の運動方程式を積分して，系の絶対加速度応答，相対速度応答お
よび相対変位応答時刻歴を計算するとともに，併せてこれら応答の最大値を求めるサブルーチン
副プログラムである． 
 
 

RESP（1 質点減衰系の地震応答） 
 
【目 的】 
 与えられた固有円振動数および減衰定数をもつ 1 質点系の，与えられた地動加速度時刻歴に対
する絶対加速度応答，相対速度応答および相対変位応答時刻歴ならびにそれぞれの最大値を計算
する. 
 
【使用法】 
（1） 接続方法 
 CALL RESP (H，W，DT，NN，DDY，ACC，VEL,  DIS,  ND， SA，SV,  SD) 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

引　数 型 プログラムを呼ぶときの内容 プログラムから戻ったときの内容

H R  減衰定数 (無次元小数)  不変

W R  固有円振動数 (単位 rad/sec)  不変

DT R  時刻歴の時間間隔 (単位 sec)  不変

NN I  時刻歴のデータ総数  不変

DDY
R

1次元配列

（ND）
 地動加速度時刻歴 (単位 Gal)  不変

ACC
R

1次元配列
（ND）

 何も入れなくてよい  絶対加速度応答時刻歴 (単位 Gal)

VEL
R

1次元配列
（ND）

 何も入れなくてよい  相対速度応答時刻歴 (単位 kine)

DIS
R

1次元配列
（ND）

 何も入れなくてよい  相対変位応答時刻歴 (単位 cm)

ND I  主プログラムにおけるDDY, ACC,
 VEL, DISのディメンション

 不変 

（2） 必要なサブルーチン及び関数副プログラム 

ない 
 

【プログラム・リスト】 
 
C  * * * * * * * * * * * * * * * * * * * * * * * * * *                  RESP   1 

C     SUBROUTINE FOR RESPONSE OF SINGLE-DOF SYSTEM                      RESP   2 

C  * * * * * * * * * * * * * * * * * * * * * * * * * *                  RESP   3 

C                                                                       RESP   4 

C                                CODED BY Y.OHSAKI                      RESP   5 

C                                                                       RESP   6 

C     PURPOSE                                                           RESP   7 

C       TO COMPUTE ABSOLUTE ACCELERATION, RELATIVE VELOCITY AND RELA-   RESP   8 

C       TIVE DISPLACEMENT RESPONSE TIME HISTORIES AND THEIR MAXIMA OF   RESP   9 

C       A SINGLE DEGREE-OF-FREEDOM SYSTEM WITH GIVEN NATURAL FREQUENCY  RESP  10 

C       AND DAMPING FACTOR EXCITED BY GIVEN ACCELERATIONS               RESP  11 

C                                                                       RESP  12 

C     USAGE                                                             RESP  13 

C       CALL RESP(H,W,DT,NN,DDY,ACC,VEL,DIS,ND,SA,SV,SD)                RESP  14 

C                                                                       RESP  15 

C     DESCRIPTION OF ARGUMENTS                                          RESP  16 

C       H       - DAMPING FACTOR IN DECIMAL FRACTION                    RESP  17 

C       W       - NATURAL CIRCULAR FREQUENCY IN RAD/SEC                 RESP  18 

C       DT      - TIME INCREMENT IN TIME HISTORIES IN SEC               RESP  19 

C       NN      - TOTAL NUMBER OF DATA IN TIME HISTORIES                RESP  20 

C       DDY(ND) - GIVEN ACCELERATION TIME HISTORY IN GALS               RESP  21 

C       ACC(ND) - ABSOLUTE ACCELERATION RESPONSE TIME HISTORY IN GALS   RESP  22 

C       VEL(ND) - RELATIVE VELOCITY RESPONSE TIME HISTORY IN KINES      RESP  23 

C       DIS(ND) - RELATIVE DISPLACEMENT RESPONSE TIME HISTORY IN CENTI- RESP  24 

C                 METERS                                                RESP  25 

C       ND      - DIMENSION OF DDY,ACC,VEL,DIS IN CALLING PROGRAM       RESP  26 

C       SA      - MAX. ABSOLUTE ACCELERATION RESPONSE IN GALS           RESP  27 

C       SV      - MAX. RELATIVE VELOCITY RESPONSE IN KINES              RESP  28 

C       SD      - MAX. RELATIVE DISPLACEMENT RESPONSE IN CENTIMETERS    RESP  29 

C                                                                       RESP  30 

C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED                     RESP  31 

C       NONE                                                            RESP  32 

C                                                                       RESP  33 

      SUBROUTINE RESP(H,W,DT,NN,DDY,ACC,VEL,DIS,ND,SA,SV,SD)            RESP  34 

C                                                                       RESP  35 

      DIMENSION DDY(ND),ACC(ND),VEL(ND),DIS(ND)                         RESP  36 

C                                                                       RESP  37 

      W2=W*W                                                            RESP  38 

      HW=H*W                                                            RESP  39 

      WD=W*SQRT(1.-H*H)                                                 RESP  40 

SA R  何も入れなくてよい  最大絶対加速度応答 (単位 Gal)

SV R  何も入れなくてよい  最大相対速度応答 (単位 kine)

SD R  何も入れなくてよい  最大相対変位応答 (単位 cm) 

      WDT=WD*DT                                                         RESP  41 

      E=EXP(-HW*DT)                                                     RESP  42 

      CWDT=COS(WDT)                                                     RESP  43 

      SWDT=SIN(WDT)                                                     RESP  44 

      A11= E*(CWDT+HW*SWDT/WD)                                          RESP  45 

      A12= E*SWDT/WD                                                    RESP  46 

      A21=-E*W2*SWDT/WD                                                 RESP  47 

      A22= E*(CWDT-HW*SWDT/WD)                                          RESP  48 

      SS=-HW*SWDT-WD*CWDT                                               RESP  49 

      CC=-HW*CWDT+WD*SWDT                                               RESP  50 

      S1=(E*SS+WD)/W2                                                   RESP  51 

      C1=(E*CC+HW)/W2                                                   RESP  52 

      S2=(E*DT*SS+HW*S1+WD*C1)/W2                                       RESP  53 

      C2=(E*DT*CC+HW*C1-WD*S1)/W2                                       RESP  54 

      S3=DT*S1-S2                                                       RESP  55 

      C3=DT*C1-C2                                                       RESP  56 

      B11=-S2/WDT                                                       RESP  57 

      B12=-S3/WDT                                                       RESP  58 

      B21=(HW*S2-WD*C2)/WDT                                             RESP  59 

      B22=(HW*S3-WD*C3)/WDT                                             RESP  60 

      ACC(1)=2.*H*W*DDY(1)*DT                                           RESP  61 

      VEL(1)=-DDY(1)*DT                                                 RESP  62 

      DIS(1)=0.                                                         RESP  63 

      DX=VEL(1)                                                         RESP  64 

      X=0.                                                              RESP  65 

      SA=0.                                                             RESP  66 

      SV=0.                                                             RESP  67 

      SD=0.                                                             RESP  68 

      DO 110 M=2,NN                                                     RESP  69 

      DXF=DX                                                            RESP  70 

      XF=X                                                              RESP  71 

      DDYM=DDY(M)                                                       RESP  72 

      DDYF=DDY(M-1)                                                     RESP  73 

      X= A12*DXF+A11*XF+B12*DDYM+B11*DDYF                               RESP  74 

      DX=A22*DXF+A21*XF+B22*DDYM+B21*DDYF                               RESP  75 

      DDX=-2.*HW*DX-W2*X                                                RESP  76 

      ACC(M)=DDX                                                        RESP  77 

      VEL(M)=DX                                                         RESP  78 

      DIS(M)=X                                                          RESP  79 

      SA=AMAX1(SA,ABS(DDX))                                             RESP  80 

      SV=AMAX1(SV,ABS(DX))                                              RESP  81 

      SD=AMAX1(SD,ABS(X))                                               RESP  82 

  110 CONTINUE                                                          RESP  83 

      RETURN                                                            RESP  84 

      END                                                               RESP  85 

        

 

 

【使用例】 
 非減衰固有周期=0.3 sec．減衰定数 h=5%の 1 質点減衰系に，エル・セントロ地震波（EQ.01）
の加速度時刻歴を，地動として与えた場合の絶対加速度応答，相対速度応答および相対変位応答

を計算する．結果は，それらがそれぞれ配列 ACC，VEL，DIS に格納されており，図示すれば
以下のようになる． 
 

      DIMENSION DDY(800),ACC(800),VEL(800),DIS(800) 

      DATA H/0.05/,T/0.3/ 

C 

      READ(50,501) DT,NN,(DDY(M),M=1,NN) 

      W=6.283185/T 

      CALL RESP (H,W,DT,NN,DDY,ACC,VEL,DIS,800,SA,SV,SD) 

      STOP 

C 

  501 FORMAT(T51,F10.0,I10/(8F10.0))     

      END       

 

 

アウトプット: 

相対変位応答（最大 1.77 cm)


